<commit_before>{ lib, stdenv, fetchFromGitHub, makeWrapper
, adoptopenjdk-bin, jre, ant
}:

stdenv.mkDerivation rec {
  pname = "tlaplus";
  version = "1.7.0";

  src = fetchFromGitHub {
    owner  = "tlaplus";
    repo   = "tlaplus";
    rev    = "refs/tags/v${version}";
    sha256 = "1mm6r9bq79zks50yk0agcpdkw9yy994m38ibmgpb3bi3wkpq9891";
  };

  nativeBuildInputs = [ makeWrapper ];
  buildInputs = [ adoptopenjdk-bin ant ];

  buildPhase = "ant -f tlatools/org.lamport.tlatools/customBuild.xml compile dist";
  installPhase = ''
    mkdir -p $out/share/java $out/bin
    cp tlatools/org.lamport.tlatools/dist/*.jar $out/share/java

    makeWrapper ${jre}/bin/java $out/bin/tlc2 \
      --add-flags "-cp $out/share/java/tla2tools.jar tlc2.TLC"
    makeWrapper ${jre}/bin/java $out/bin/tla2sany \
      --add-flags "-cp $out/share/java/tla2tools.jar tla2sany.SANY"
    makeWrapper ${jre}/bin/java $out/bin/pcal \
      --add-flags "-cp $out/share/java/tla2tools.jar pcal.trans"
    makeWrapper ${jre}/bin/java $out/bin/tla2tex \
      --add-flags "-cp $out/share/java/tla2tools.jar tla2tex.TLA"
  '';

  meta = {
    description = "An algorithm specification language with model checking tools";
    homepage    = "http://lamport.azurewebsites.net/tla/tla.html";
    license     = lib.licenses.mit;
    platforms   = lib.platforms.unix;
    maintainers = [ lib.maintainers.thoughtpolice ];
  };
}<commit_msg>tlaplus: 1.7.0 -> 1.7.1<commit_after>{ lib, stdenv, fetchurl, makeWrapper, adoptopenjdk-bin, jre }:

stdenv.mkDerivation rec {
  pname = "tlaplus";
  version = "1.7.1";

  src = fetchurl {
    url = "https://github.com/tlaplus/tlaplus/releases/download/v${version}/tla2tools.jar";
    sha256 = "d532ba31aafe17afba1130f92410d9257454ff7393d1eb2fe032f0c07f352da5";
  };

  nativeBuildInputs = [ makeWrapper ];
  buildInputs = [ adoptopenjdk-bin ];

  dontUnpack = true;
  installPhase = ''
    mkdir -p $out/share/java $out/bin
    cp $src $out/share/java/tla2tools.jar

    makeWrapper ${jre}/bin/java $out/bin/tlc \
      --add-flags "-XX:+UseParallelGC -cp $out/share/java/tla2tools.jar tlc2.TLC"
    makeWrapper ${jre}/bin/java $out/bin/tlasany \
      --add-flags "-XX:+UseParallelGC -cp $out/share/java/tla2tools.jar tla2sany.SANY"
    makeWrapper ${jre}/bin/java $out/bin/pcal \
      --add-flags "-XX:+UseParallelGC -cp $out/share/java/tla2tools.jar pcal.trans"
    makeWrapper ${jre}/bin/java $out/bin/tlatex \
      --add-flags "-XX:+UseParallelGC -cp $out/share/java/tla2tools.jar tla2tex.TLA"
  '';

  meta = {
    description = "An algorithm specification language with model checking tools";
    homepage    = "http://lamport.azurewebsites.net/tla/tla.html";
    license     = lib.licenses.mit;
    platforms   = lib.platforms.unix;
    maintainers = with lib.maintainers; [ florentc thoughtpolice ];
  };
}
{'fraction_non_alphanumeric': 0.15093672675857192, 'fraction_numerical': 0.033934252386002124, 'mean_word_length': 3.1495601173020527, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 1, 'lorem ipsum': 0, 'www.': 0, 'xml': 1}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 1, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6714614', 'n_tokens_mistral': 1159, 'n_tokens_neox': 1099, 'n_words': 177}